Clifford deformation of code Hamiltonians

This *Mathematica* script provides a workspace for some computations with translationally invariant code Hamiltonians. They are known as stabilizer Hamiltonians --- frustration-free Hamiltonians whose terms, consisting of tensor product of Pauli matrices, commute with any other term.

A traslationally invariant stabilizer (code) Hamiltonian on a D-dimensional lattice is represented by a Laurent polynomial matrix of size 2q×t in D variables. The lattice is just $\mathbb{Z}^D$, and for each lattice site there lie q qubits. t is the number of distinct types of stabilizer generators. In other words, t is the number of strictly different terms of the Hamiltonian up to translations. For example, the classical 2D Ising model on the square lattice has t = 2; one for the horizontal coupling and another for the veritcal coupling. For the toric code Hamiltonian on the square lattice, t is also equal to 2; one for the stars and another for the plaquettes.

Under conjugations by Clifford operators, the terms in the Hamiltonian are mapped to some other Pauli operator, by definiton of Clifford operators. We only consider a translation-invariant product of local Clifford operators of non-overlapping support. Hence, the conjugated Hamiltonian (or code) is representated by a new Laurent polynomial matrix. The script below makes this transformation explicit and convenient.

Below, entanglement renormalization for the toric codes (Ising gauge theory) and the cubic code is presented using Clifford transformations. For the toric codes in 2, 3, and 4 spatial dimensions, the calculation covers not only the $\mathbb{Z}_2$ toric code, but also $\mathbb{Z}_p$ toric codes with prime number p. On the other hand, the cubic code calculation depends on the base field $\mathbb{Z}_2 = \mathbb{F}_2$. Evaluating all cells in this notebook may take a few seconds. Each section will be expanded by double-clicking a small triangle on the far right.

27 January 2014
Jeongwan Haah

Tools

All routines here do not depend on the characteristic of the base field. So they are applicable for stabilizer codes on qudits of prime dimensions as well. One has to change the characteristic 2 in ``display'' function below to some prime number. Make sure not to use rational numbers in the intermediate calculation since *Mathematica* does not automatically carry out modulo computations.

The naming policy for functions is such that the first letter is in lower case in order to distinguish it from *Mathematica*'s own functions.

Variables of Laurent polynomial ring:

In[1]:=
```
translationVariables = {x, y, z, w};
```

display(mat)

In[2]:=
```
display[mat_] := MatrixForm[Expand[mat]]
display2[mat_] := MatrixForm[Expand[PolynomialMod[mat, 2]]] (*for qubit-stabilizer codes*)
```

coarseGrain(mat, {var, n})

In[4]:=
```
expandEntry[poly_, x_, n_] := Module[{f, p, c, xp, col, monos, exps, gen},
  f[i_, j_] := If[i - 1 == j, 1, If[i - 1 + n == j, x, 0]];
  gen = Table[f[i, j], {i, 1, n}, {j, 1, n}];
  p = Expand[poly];
  If[p === 0,
   (*return zero matrix if the input is zero*)
   Table[0, {n}, {n}],
   (*Else, compute properly*)
   col = Collect[p, x];
   monos = If[Head[col] === Plus, Apply[List, col], {col}];
   exps = Map[Exponent[#, x] &, monos];
   c = Power[x, -exps] * monos;
   xp = Map[MatrixPower[gen, #] &, exps];
   Expand[c.xp]
   (*end-if*)]
 ]
```

In[5]:=
```
coarseGrain[mat_, {x_, n_}] :=
 Apply[Join, Table[
   Join[Table[
     Flatten[
      Map[expandEntry[#, x, n][[nn]] &, mat[[row]]]
     ], {nn, n}]
   ], {row, Length[mat]}]
 ]
```

In[6]:=
```
coarseGrainReverseBlockOrder[mat_, {x_, n_}] := Module[{revMat, twoq, t},
  revMat = Reverse[IdentityMatrix[n]];
  {twoq, t} = Dimensions[mat];
  KroneckerProduct[IdentityMatrix[twoq], revMat].
   coarseGrain[mat, {x, n}].KroneckerProduct[IdentityMatrix[t], revMat]
 ]
```

The function **coarseGrain** sorts the qubits in the following order upon blocking n sites along x direction as one new site: If {1,2,...,q} were labels of the qubits at an old site, then nq qubits are ordered as {1 at x^0, 1 at x^1,... 1 at x^{n-1}, 2 at x^0, 2 at x^1,..., 2 at x^{n-1}, q at x^0, q at x^1,..., q at x^{n-1}}. The last *Mathematica* function **coarseGrainReverseBlockOrder** sorts them as {1 at x^{n-1}, 1 at x^{n-2},..., 1 at x^0, 2 at x^{n-1},..., 2 at x^0,...,q at x^{n-1}, q at x^0}.

antipode(f): an involutive automorphism of Laurent polynomial ring
dagger = antipode followed by transpose
symprod[v, w] = symplectic product = dagger[v].λ.w

In[7]:=
```
monomialAntipode[f_] :=
 Module[{e, v, c, mono},
  v = Intersection[Variables[f], translationVariables];
  e = Map[Exponent[f, #] &, v];
  mono = Apply[Times, Power[v, e]];
  c = If[mono === 1, f, Coefficient[f, mono]];
  c * Apply[Times, Power[v, -e]]]
antipode[f_] :=
 Module[{ff, fff},
  ff = Expand[f];
  If[Head[ff] === Plus, fff = ff, fff = {ff}];
  Apply[Plus, Map[monomialAntipode, Apply[List, Expand[fff]]]]
 ]
```

In[9]:=
```
dagger[m_] := Transpose[Map[antipode, m, {ArrayDepth[m]}]]
```

In[10]:=
```
symprod[v_, w_] := Expand[dagger[v].KroneckerProduct[({{0, 1}, {-1, 0}}), IdentityMatrix[Length[w] / 2]].w]
```

Elementary Symplectic Transformations

hadamard(q, i-th)

hadamard(q, {i-th})

cNot(q, {tgt, src, f})

cPhase(q, {tgt, f})

In cNot, f can be any Laurent polynomial. However, in cPhase, f must be self-antipode; otherwise, it won’t be symplectic.

In[11]:=
```
hadamard[mat_, i_] :=
 (*`i' can either be a number or a list of numbers*)
 If[Head[i] === List,
  Fold[hadamard, mat, i],
  Module[{m, q},
   q = Length[mat] / 2;
   m = IdentityMatrix[2 q];
   m[[i, i]] = 0;
   m[[q + i, q + i]] = 0;
   m[[q + i, i]] = 1;
   m[[i, q + i]] = -1;
   m.mat
  ]
 ]
```

In[12]:=
```
cNotMatrix[q_, tgt_, src_, f_] := Module[{m},
  m = IdentityMatrix[2 q];
  m[[tgt, src]] = f;
  m[[src + q, tgt + q]] = -antipode[f];
  m
 ]
cNot[mat_, {tgt_, src_, f_}] := cNotMatrix[Length[mat] / 2, tgt, src, f].mat
```

```
In[14]:= cPhaseMatrix[q_, tgt_, f_] := Module[{m},
    m = IdentityMatrix[2 q];
    m[[tgt + q, tgt]] = f;
    m
   ]
  cPhase[mat_, {tgt_, f_}] := cPhaseMatrix[Length[mat] / 2, tgt, f].mat
```

Elementary column operation: colOp(mat, {tgt, src, f})

```
In[16]:= colOpMatrix[t_, tgt_, src_, f_] := Module[{m},
    m = IdentityMatrix[t];
    m[[src, tgt]] = f;
    m
   ]
  colOp[mat_, {tgt_, src_, f_}] := mat.colOpMatrix[Dimensions[mat][[2]], tgt, src, f]
```

Factoring out ancillas:
factorOut(matrix, tgt or {tgts}) --- dropping (ancillary) qubit(s)
dropColumn(matrix, tgt or {tgts}) --- dropping column(s)

```
In[18]:= factorOut[mat_, tgt_] :=
   (*`tgt' can either be a number or a list of numbers*)
   Fold[Delete, mat, Reverse[Sort[Flatten[{{tgt}, {tgt + Length[mat] / 2}}]]]]

In[19]:= dropColumn[mat_, cols_] :=
   (*`cols' can either be a number or a list of numbers*)
   Transpose[Fold[Delete, Transpose[mat], Reverse[Sort[Flatten[{cols}]]]]]
```

Short examples

In[20]:= antipode$\left[-31 + 4321\, x\, y\, (z + 1) - 432 + z^{-1} + 123\, y^{13} / z\right]$

Out[20]= $-463 + \frac{4321}{x\, y} + \frac{4321}{x\, y\, z} + z + \frac{123\, z}{y^{13}}$

In[21]:=
```
cNotMatrix[2, 1, 2, x y + 1 + z^-1] // MatrixForm
```

Out[21]//MatrixForm=

$$\begin{pmatrix} 1 & 1 + x\,y + \frac{1}{z} & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & -1 - \frac{1}{x\,y} - z & 1 \end{pmatrix}$$

In[22]:=
```
cPhaseMatrix[2, 2, 1] // MatrixForm
```

Out[22]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 1 \end{pmatrix}$$

In[23]:=
```
antipode[6 x + 3 s]
antipode[antipode[6 x + 3 s]]
```

Out[23]= $3\,s + \frac{6}{x}$

Out[24]= $3\,s + 6\,x$

Note on Mathematica syntax

The standard syntax for Mathematica functions is like **SomeFunction[argument]**. When there are two arguments, the following are equivalent: **SomeFunction2[arg1,arg2]** and **arg1~SomeFunction2~arg2.** The latter expression is particularly handy in the following calculations. The latter expression is evaluated from the beginning.

2D toric code

In[25]:=

$$\text{sigmaToric} = \begin{pmatrix} x - 1 & 0 \\ y - 1 & 0 \\ 0 & -y^{-1} + 1 \\ 0 & x^{-1} - 1 \end{pmatrix};$$

In[26]:=
```
sigmaToric // dagger // display
```

Out[26]//MatrixForm=

$$\begin{pmatrix} -1+\frac{1}{x} & -1+\frac{1}{y} & 0 & 0 \\ 0 & 0 & 1-y & -1+x \end{pmatrix}$$

In[27]:=
```
symprod[sigmaToric, sigmaToric]
```

Out[27]= {{0, 0}, {0, 0}}

In[28]:=
```
sigmaToric1 = sigmaToric\
   ~coarseGrainReverseBlockOrder~{x, 2} (*blocking 2 nearby qubits in x-direction*)\
   ~cNot~{2, 1, x}\
   ~cNot~{3, 1, y - 1}\
   ~cNot~{4, 3, -1}\
   ~colOp~{2, 1, 1}\
   ~colOp~{4, 3, 1};
display[sigmaToric1]
```

Out[29]//MatrixForm=

$$\begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & -1+x & 0 & 0 \\ 0 & -1+y & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1-\frac{1}{y} \\ 0 & 0 & 0 & -1+\frac{1}{x} \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

In[30]:=
```
sigmaToric1\
  ~factorOut~{1, 4}~dropColumn~{1, 3}\
  // display
```

Out[30]//MatrixForm=

$$\begin{pmatrix} -1+x & 0 \\ -1+y & 0 \\ 0 & 1-\frac{1}{y} \\ 0 & -1+\frac{1}{x} \end{pmatrix}$$

3D toric code

In[31]:=
$$\text{sigma3t} = \begin{pmatrix} 1 - x^{-1} & 0 & 0 & 0 \\ 1 - y^{-1} & 0 & 0 & 0 \\ 1 - z^{-1} & 0 & 0 & 0 \\ 0 & 0 & -z+1 & -y+1 \\ 0 & -z+1 & 0 & x-1 \\ 0 & y-1 & x-1 & 0 \end{pmatrix};$$

In[32]:=
```
symprod[sigma3t, sigma3t]
```

Out[32]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

In[33]:=
```
sigma3t1 = sigma3t\
    ~coarseGrainReverseBlockOrder~{x, 2}\
    ~cNot~{2, 1, 1}~cNot~{3, 1, -1 + 1 / y}~cNot~{5, 1, -1 + 1 / z}\
    ~cNot~{3, 4, -1 / x}~cNot~{5, 6, -1 / x}\
    ~colOp~{2, 1, 1 / x}~colOp~{3, 5, -1 + y}~colOp~{6, 5, 1}~colOp~{8, 7, 1}~colOp~{3, 7, 1 - z}~colOp~{3, 4, -x};
sigma3t1 // display
```

Out[34]//MatrixForm=
$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1-\frac{1}{x} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1-\frac{1}{y} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1-\frac{1}{z} & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1-z & 0 & 1-y \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1-z & 0 & 0 & 0 & -1+x \\ 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1+y & 0 & -1+x & 0 & 0 \end{pmatrix}$$

In[35]:=
```
sigma3t1~factorOut~{1, 5, 3}~dropColumn~{1, 3, 5, 7} // display
```

Out[35]//MatrixForm=

$$\begin{pmatrix} 1-\frac{1}{x} & 0 & 0 & 0 \\ 1-\frac{1}{y} & 0 & 0 & 0 \\ 1-\frac{1}{z} & 0 & 0 & 0 \\ 0 & 0 & 1-z & 1-y \\ 0 & 1-z & 0 & -1+x \\ 0 & -1+y & -1+x & 0 \end{pmatrix}$$

4D toric code

In[36]:=

$$\text{s4a} = \begin{pmatrix} y & -x & 0 & 0 \\ z & 0 & -x & 0 \\ w & 0 & 0 & -x \\ 0 & z & -y & 0 \\ 0 & w & 0 & -y \\ 0 & 0 & w & -z \end{pmatrix};$$

In[37]:=
```
Transpose[s4a] // display
```

Out[37]//MatrixForm=

$$\begin{pmatrix} y & z & w & 0 & 0 & 0 \\ -x & 0 & 0 & z & w & 0 \\ 0 & -x & 0 & -y & 0 & w \\ 0 & 0 & -x & 0 & -y & -z \end{pmatrix}$$

In[38]:=

$$\text{s4b} = \begin{pmatrix} 0 & 0 & z & w \\ 0 & w & -y & 0 \\ 0 & -z & 0 & -y \\ w & 0 & x & 0 \\ -z & 0 & 0 & x \\ y & x & 0 & 0 \end{pmatrix};$$

The matrices are obtained from cellular homology induced from simple cubic lattice filling 4 - space. The rows of s4a generate the kernel of the map $(x \; y \; z \; w)$. s4b generate the kernel of the transpose of s4a.

In[39]:=
```
sigma4t = KroneckerProduct[(1 0
                             0 0), s4a /. {x → x - 1, y → y - 1, z → z - 1, w → w - 1}] +
   KroneckerProduct[(0 0
                     0 1), Transpose[dagger[s4b /. {x → x - 1, y → y - 1, z → z - 1, w → w - 1}]]];
sigma4t //
 display
```

Out[40]//MatrixForm=

$$\begin{pmatrix}
-1+y & 1-x & 0 & 0 & 0 & 0 & 0 & 0 \\
-1+z & 0 & 1-x & 0 & 0 & 0 & 0 & 0 \\
-1+w & 0 & 0 & 1-x & 0 & 0 & 0 & 0 \\
0 & -1+z & 1-y & 0 & 0 & 0 & 0 & 0 \\
0 & -1+w & 0 & 1-y & 0 & 0 & 0 & 0 \\
0 & 0 & -1+w & 1-z & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{z} & -1+\frac{1}{w} \\
0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & 1-\frac{1}{y} & 0 \\
0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 1-\frac{1}{y} \\
0 & 0 & 0 & 0 & -1+\frac{1}{w} & 0 & -1+\frac{1}{x} & 0 \\
0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 0 & -1+\frac{1}{x} \\
0 & 0 & 0 & 0 & -1+\frac{1}{y} & -1+\frac{1}{x} & 0 & 0
\end{pmatrix}$$

In[41]:=
```
symprod[sigma4t, sigma4t]
```

Out[41]=
```
{{0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0},
 {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}, {0, 0, 0, 0, 0, 0, 0, 0}}
```

In[42]:=
```
sigma4t1 = sigma4t\
   ~coarseGrainReverseBlockOrder~{y, 2}\
   ~cNot~{2, 1, y}~cNot~{3, 1, z - 1}~cNot~{5, 1, w - 1}\
   ~cNot~{3, 7, x - 1}~cNot~{8, 7, y}~cNot~{11, 7, -w + 1}\
   ~cNot~{9, 10, 1}~cNot~{6, 10, x - 1}~cNot~{12, 10, z - 1}\
   ~cNot~{4, 3, -1}\
   ~colOp~{2, 1, 1}~colOp~{3, 1, 1 - x}~colOp~{14, 13, 1}~colOp~
   {12, 13, 1 / w - 1}~colOp~{3, 5, 1 - z}~colOp~{6, 5, 1}~colOp~{7, 8, y}~colOp~{4, 8, 1 - w};
sigma4t1 //
 display
```

Out[43]//MatrixForm=

$$\begin{pmatrix}
-1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+y & y-xy & 1-x & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+z & 0 & 0 & 0 & 1-x & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+w & -1+w+x-wx & 0 & 0 & 0 & 1-x & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+w & 0 & 1-w-x+wx & 0 & 0 & y-xy & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -y+yz & -1+z & 0 & 1-y & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -1+w & -1+w & 0 & 0 & 1-y & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -1+w+z-wz & 0 & 0 & -1+w & 1-z & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1-w-z+wz & 0 & -1+w & y-yz & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{z} & 0 & -1+\frac{1}{w} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & -1+\frac{1}{w} & 0 & 1-\frac{1}{y} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 0 & 0 & 1 & -\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 0 & -1 & 1 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & 0 & 0 & 0 & -1+\frac{1}{x} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{x} & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & \frac{1}{y} & -1+\frac{1}{x} & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & -1 & 0 & -1+\frac{1}{x} & 0 & 0 & 0 & 0
\end{pmatrix}$$

In[44]:=
```
sigma4t2 = sigma4t1\
  ~factorOut~{1, 4, 7, 10}~dropColumn~{1, 5, 13, 8}\
  ~cNot~{3, 4, -1}~cNot~{3, 6, x - 1}~colOp~{8, 11, -1 + 1 / z}~colOp~{12, 11, 1 / y}\
  ~cNot~{8, 7, -1}~cNot~{8, 6, 1 - z}~colOp~{7, 6, 1}~colOp~{9, 6, 1 - 1 / x};
sigma4t2 // display
```

Out[45]//MatrixForm=

$$\begin{pmatrix}
-1+y & y-xy & 1-x & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
-1+z & 0 & 0 & 1-x & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
-1+w & 0 & 1-w-x+wx & 0 & y-xy & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -y+yz & -1+z & 1-y & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+w & -1+w & 0 & 1-y & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+w+z-wz & 0 & -1+w & 1-z & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{z} & 0 & -1+\frac{1}{w} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & -1+\frac{1}{w} & 1-\frac{1}{y} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 1-\frac{1}{z} & 0 & 0 & 1-\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & 0 & 0 & -1+\frac{1}{x} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 1-\frac{1}{x}-\frac{1}{z}+\frac{1}{xz} & 1-\frac{1}{x}-\frac{1}{z}+\frac{1}{xz} & 0 & 0 & -\frac{1}{y}+\frac{1}{xy} \\
0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{y} & -1+\frac{1}{x} & -1+\frac{1}{x} & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0
\end{pmatrix}$$

In[46]:=
```
sigma4t3 = sigma4t2\
    ~factorOut~{3, 8}~dropColumn~{6, 11}\
    ~cNot~{6, 5, z}~cNot~{6, 5, -z}~cNot~{3, 5, 1 - x}\
    ~colOp~{2, 3, -y}~colOp~{2, 5, 1 - w}~colOp~{8, 7, -1};
sigma4t3 // display
```

Out[47]//MatrixForm=

$$\begin{pmatrix}
-1+y & 0 & 1-x & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
-1+z & 0 & 0 & 1-x & 0 & 0 & 0 & 0 & 0 & 0 \\
-1+w & 0 & 0 & 0 & 1-x & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -1+z & 1-y & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -1+w & 0 & 1-y & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -1+w & 1-z & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{z} & -1+\frac{1}{w} \\
0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & 0 & 1-\frac{1}{y} & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 0 & 1-\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & -1+\frac{1}{w} & 0 & 0 & -1+\frac{1}{x} & 0 \\
0 & 0 & 0 & 0 & 0 & 1-\frac{1}{z} & 0 & 0 & 0 & -1+\frac{1}{x} \\
0 & 0 & 0 & 0 & 0 & -1+\frac{1}{y} & -1+\frac{1}{x} & 0 & 0 & 0
\end{pmatrix}$$

In[48]:=
```
Expand[sigma4t3~dropColumn~{2, 8}] == sigma4t
```

Out[48]= True

Cubic code

In[49]:=

$$\texttt{sigmaCubic} = \begin{pmatrix}
1+x+y+z & 0 \\
1+x\,y+y\,z+z\,x & 0 \\
0 & 1+x^{-1}\,y^{-1}+y^{-1}\,z^{-1}+z^{-1}\,x^{-1} \\
0 & 1+x^{-1}+y^{-1}+z^{-1}
\end{pmatrix};$$

In[50]:=
```
sigmaCubic1 =
  sigmaCubic\
   ~coarseGrainReverseBlockOrder~{x, 2}\
   ~colOp~{1, 2, -y - z - 1}\
   ~colOp~{4, 3, -(z y + y + z)/(y z)}\
   ~cNot~{2, 1, -(y + z + 1)}\
   ~cNot~{3, 1, -y - z}\
   ~cNot~{4, 1, -1 - y z}\
   ~cNot~{4, 3, z + 1 + y}\
   ~cNot~{4, 2, z + y};
sigmaCubic1 // display2
```

Out[51]//MatrixForm=

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 1+x+y^2+z^2 & 0 & 0 & 0 \\ 1+y+y^2+z+y\,z+z^2 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1+\frac{1}{y^2}+\frac{1}{y}+\frac{1}{z^2}+\frac{1}{z}+\frac{1}{y\,z} \\ 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y^2}+\frac{1}{z^2} \\ 0 & 0 & 1 & 0 \end{pmatrix}$$

In[52]:=
```
sigmaCubic2 =
  sigmaCubic1\
   ~factorOut~{1, 4}~dropColumn~{2, 3};
display2[sigmaCubic2]
```

Out[53]//MatrixForm=

$$\begin{pmatrix} 1+x+y^2+z^2 & 0 \\ 1+y+y^2+z+y\,z+z^2 & 0 \\ 0 & 1+\frac{1}{y^2}+\frac{1}{y}+\frac{1}{z^2}+\frac{1}{z}+\frac{1}{y\,z} \\ 0 & 1+\frac{1}{x}+\frac{1}{y^2}+\frac{1}{z^2} \end{pmatrix}$$

In[54]:=
```
sigmaCubic3 =
  sigmaCubic2\
   ~coarseGrainReverseBlockOrder~{y, 2}\
   ~cNot~{1, 3, -1}\
   ~coarseGrainReverseBlockOrder~{z, 2}\
   ~colOp~{1, 2, -x}~colOp~{3, 2, -1}~colOp~{4, 2, -1}\
   ~cNot~{2, 1, -x}~cNot~{5, 1, -1}~cNot~{6, 1, -y - z - 1}\
   ~cNot~{7, 1, -y}~cNot~{8, 1, -y}\
   ~colOp~{6, 5, -1 - z^-1 - y^-1}~colOp~{7, 5, -y^-1}~colOp~{8, 5, -y^-1}\
   ~cNot~{2, 3, 1}~cNot~{2, 4, 1}~cNot~{2, 5, x}~cNot~{2, 6, 1};
sigmaCubic3 // display2
```

Out[55]//MatrixForm=

$$\begin{pmatrix}
0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 1+x+y+z & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1+x+y+z & 0 & 0 & 0 & 0 \\
1+x+y+z & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
x+xy+z+xz & 0 & 1+y & y+z & 0 & 0 & 0 & 0 \\
y+xy & 0 & 1+z & 1+y & 0 & 0 & 0 & 0 \\
xy+yz & 0 & y+z & 1+z & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{y} & 1+\frac{1}{z} & \frac{1}{y}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{y}+\frac{1}{z} & 1+\frac{1}{y} & 1+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{x}+\frac{1}{xy}+\frac{1}{z}+\frac{1}{xz} & \frac{1}{y}+\frac{1}{xy} & \frac{1}{xy}+\frac{1}{yz} \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z}
\end{pmatrix}$$

In[56]:=
```
sigmaCubic4 =
  sigmaCubic3\
   ~factorOut~{1, 2}~dropColumn~{2, 5}\
   ~cNot~{6, 4, -1}~cNot~{6, 5, -1}~cNot~{6, 3, -1}~cNot~{4, 3, -1}~cNot~{4, 5, -1}\
   ~colOp~{1, 3, -x}~cNot~{2, 3, x}~colOp~{1, 2, 1}\
   ~cNot~{3, 1, -1}~cNot~{4, 1, -1}~cNot~{5, 1, -1}~cNot~{1, 3, 1}\
   ~cNot~{2, 6, -1}~cNot~{6, 2, 1}~cNot~{2, 6, -1} (* Permute qubit 2,6 *)\
   ~colOp~{6, 4, -1}~colOp~{6, 5, 1}\
   ~cNot~{6, 4, 1}~cNot~{6, 5, 1}~cNot~{3, 6, 1};
sigmaCubic4 // display2
```

Out[57]//MatrixForm=

$$\begin{pmatrix} 1+x+y+z & 0 & 0 & 0 & 0 & 0 \\ 1+xy+xz+yz & 0 & 0 & 0 & 0 & 0 \\ 0 & x+z & 1+x & 0 & 0 & 0 \\ 0 & 1+x & 1+z & 0 & 0 & 0 \\ 0 & x+y & 1+y & 0 & 0 & 0 \\ 0 & 1+y & 1+x & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{xy}+\frac{1}{xz}+\frac{1}{yz} \\ 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} \\ 0 & 0 & 0 & 1+\frac{1}{y} & \frac{1}{x}+\frac{1}{y} & 0 \\ 0 & 0 & 0 & 1+\frac{1}{x} & 1+\frac{1}{y} & 0 \\ 0 & 0 & 0 & 1+\frac{1}{x} & \frac{1}{x}+\frac{1}{z} & 0 \\ 0 & 0 & 0 & 1+\frac{1}{z} & 1+\frac{1}{x} & 0 \end{pmatrix}$$

Here, one sees that the first two qubits (row 1, 2, 1+6, 2+6) do not interact with other qubits, and their interaction is exactly the same as the original one, sigmaCubic.
We have blocked 2^3sites into a new site, and factored $2\times 2^2 + 2=10$ qubits out that are in the trivial product state after finite number of layers of CNOT gates.

Children of cubic code

In[58]:=

$$\text{sigmaCubicB} = \begin{pmatrix} x+z & 1+x & 0 & 0 \\ 1+x & 1+z & 0 & 0 \\ x+y & 1+y & 0 & 0 \\ 1+y & 1+x & 0 & 0 \\ 0 & 0 & 1+\frac{1}{y} & \frac{1}{x}+\frac{1}{y} \\ 0 & 0 & 1+\frac{1}{x} & 1+\frac{1}{y} \\ 0 & 0 & 1+\frac{1}{x} & \frac{1}{x}+\frac{1}{z} \\ 0 & 0 & 1+\frac{1}{z} & 1+\frac{1}{x} \end{pmatrix};$$

In[59]:=

```
symprod[sigmaCubicB, sigmaCubicB] // display2
```

Out[59]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

In[60]:=
```
sigmaCubicB1 = sigmaCubicB\
   ~coarseGrainReverseBlockOrder~{x, 2}\
   ~cNot~{1, 3, z}~cNot~{2, 3, x}~cNot~{4, 3, x}~cNot~{5, 3, y}\
   ~cNot~{6, 3, x}~cNot~{7, 3, 1 + y}\
   ~colOp~{2, 1, 1}~colOp~{3, 1, 1 + z}\
   ~cNot~{4, 1, 1 + y}~cNot~{4, 5, 1}~cNot~{4, 6, 1}~cNot~{4, 7, 1 + z}\
   ~colOp~{6, 5, 1}~colOp~{8, 5, 1 + 1 / y}\
   ~cNot~{2, 1, 1}~cNot~{6, 1, 1 + y}~cNot~{7, 1, 1}~cNot~{8, 1, 1}\
   ~colOp~{2, 4, 1 + z}~colOp~{3, 4, 1 + z + z^2}\
   ~colOp~{6, 7, 1 + 1 / y}~colOp~{8, 7, 1 / y}\
   ~cNot~{2, 5, z}~cNot~{2, 6, 1}~cNot~{2, 7, 1}~cNot~{2, 8, 1}\
   ~cNot~{7, 8, 1};
sigmaCubicB1 // display2
```

Out[61]//MatrixForm=

$$\begin{pmatrix}
0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 1 + y & 1 + y z & 0 & 0 & 0 & 0 & 0 \\
0 & 1 + x + z + y z & 1 + x + y + z + x z + y z + z^2 + y z^2 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & x + y + z + y z & 0 & 0 & 0 & 0 & 0 \\
0 & y + z & 1 + x + z + z^2 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 1 + \frac{1}{x} + \frac{1}{z} + \frac{1}{y z} & 0 & 1 + \frac{1}{x} + \frac{1}{y} + \frac{1}{y z} \\
0 & 0 & 0 & 0 & 0 & 1 + \frac{1}{y} & 0 & 1 + \frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{y} + \frac{1}{z} & 0 & 1 + \frac{1}{x} + \frac{1}{z} + \frac{1}{y z} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{x} + \frac{1}{y} + \frac{1}{z} + \frac{1}{y z}
\end{pmatrix}$$

In[62]:=
```
sigmaCubicB2 = sigmaCubicB1\
   ~factorOut~{1, 2, 3, 4}~dropColumn~{1, 4, 5, 7}\
   ~coarseGrainReverseBlockOrder~{z, 2}\
   ~cNot~{2, 7, 1 + y}~cNot~{3, 7, 1 + y}~cNot~{4, 7, 1 + x}~cNot~{8, 7, y}\
   ~cNot~{2, 1, 1 + x + y + x y + z}~cNot~{3, 1, x y}~cNot~{4, 1, 1 + x x + y z}\
   ~cNot~{5, 1, x + y}~cNot~{6, 1, z + y z}~cNot~{8, 1, y + x y + z + y z}\
   ~colOp~{1, 2, y}~colOp~{3, 2, 1 + x + z}~colOp~{4, 2, 1}\
   ~colOp~{1, 3, 1 + y}~colOp~{4, 3, y}\
   ~cNot~{3, 2, y}~cNot~{3, 4, 1}~cNot~{3, 5, 1 + x}~cNot~{3, 6, 1 + y}~cNot~{3, 8, 1 + y}\
   ~colOp~{5, 7, 1 + 1 / y}~colOp~{8, 7, 1 / z};
sigmaCubicB2 // display2
```

Out[63]//MatrixForm=

$$\begin{pmatrix}
0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 \\
1+x+y+x y^2+z+y z & 0 & 0 & x y+y^2+x y^2+y z & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1+x^2+x y+x^2 y+z+y^2 z & 0 & 0 & x^2 y+z+y z+y^2 z & 0 & 0 & 0 & 0 \\
x+y+x y+y^2 & 0 & 0 & 1+y+x y+y^2 & 0 & 0 & 0 & 0 \\
z+y^2 z & 0 & 0 & x+y+y z+y^2 z & 0 & 0 & 0 & 0 \\
0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
y+x y+x y^2+y^2 z & 0 & 0 & 1+x+y+y^2+x y^2+z+y z+y^2 z & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{y^2} & 1+\frac{1}{x} & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{y z} \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{y} & 1+\frac{1}{y} & 0 & 1+\frac{1}{z} \\
0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{x y} & \frac{1}{z} & 0 & \frac{1}{x z}+\frac{1}{y z} \\
0 & 0 & 0 & 0 & \frac{1}{y^2} & \frac{1}{y} & 0 & 1+\frac{1}{x}+\frac{1}{z}+\frac{1}{y z} \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{y^2} & 0 & 0 & \frac{1}{x}+\frac{1}{y}+\frac{1}{z}+\frac{1}{y z}
\end{pmatrix}$$

In[64]:=
```
sigmaCubicB3 = sigmaCubicB2\
  ~factorOut~{1, 3, 7}~dropColumn~{2, 3, 7}\
  ~coarseGrainReverseBlockOrder~{y, 2}\
  ~cNot~{3, 1, 1 + y}~cNot~{3, 4, 1}~cNot~{3, 5, 1 + x}~cNot~{3, 6, x}\
  ~cNot~{3, 7, y}~cNot~{3, 9, 1 + y}\
  ~cNot~{8, 9, 1 + y}~cNot~{8, 7, y}~cNot~{8, 6, x}~cNot~{8, 5, 1 + x + z}\
  ~cNot~{8, 1, x + y}\
  ~colOp~{6, 5, 1 / y}~colOp~{7, 5, 1}~colOp~{8, 5, 1 / y}~colOp~{9, 5, 1 + 1 / z}\
  ~colOp~{6, 7, 1 / y}~colOp~{9, 7, 1 / z}~colOp~{10, 7, 1 + 1 / x + 1 / z};
sigmaCubicB3 // display2
```

Out[65]//MatrixForm=

$$\begin{pmatrix}
1+x+xy+z & 1+z & y+xy & x+z & 0 & 0 & 0 & 0 & 0 & 0\\
y+yz & 1+x+xy+z & xy+yz & y+xy & 0 & 0 & 0 & 0 & 0 & 0\\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0\\
xy+x^2y & 1+x^2+z+yz & x^2y+yz & z+yz & 0 & 0 & 0 & 0 & 0 & 0\\
x+y & 1+x & 1+y & 1+x & 0 & 0 & 0 & 0 & 0 & 0\\
y+xy & x+y & y+xy & 1+y & 0 & 0 & 0 & 0 & 0 & 0\\
z+yz & 0 & x+yz & 1+z & 0 & 0 & 0 & 0 & 0 & 0\\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0\\
xy+yz & 1+x & 1+x+y+xy+z+yz & 1+z & 0 & 0 & 0 & 0 & 0 & 0\\
y+xy & xy+yz & y+yz & 1+x+y+xy+z+yz & 0 & 0 & 0 & 0 & 0 & 0\\
0 & 0 & 0 & 0 & 0 & \frac{1}{y}+\frac{1}{xy} & 0 & \frac{1}{y^2}+\frac{1}{y} & \frac{1}{x}+\frac{1}{y}+\frac{1}{z}+\frac{1}{xz} & \frac{1}{x^2}+\frac{1}{x}+\frac{1}{xy}+\frac{1}{xz}\\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{y} & 0 & 1+\frac{1}{x} & 1+\frac{1}{z} & 1+\frac{1}{x}\\
0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0\\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{y} & 0 & 1+\frac{1}{y} & 1+\frac{1}{z} & 1+\frac{1}{z}\\
0 & 0 & 0 & 0 & 0 & \frac{1}{xy}+\frac{1}{yz} & 0 & \frac{1}{y}+\frac{1}{xy} & 1+\frac{1}{x}+\frac{1}{z^2}+\frac{1}{xz} & 1+\frac{1}{x^2}+\frac{1}{z^2}+\frac{1}{yz}\\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x} & 0 & \frac{1}{xy}+\frac{1}{z} & \frac{1}{x}+\frac{1}{z} & \frac{1}{x^2}+\frac{1}{x}\\
0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{y^2}+\frac{1}{y} & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} & \frac{1}{y}+\frac{1}{xy}\\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0\\
0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{1}{y^2}+\frac{1}{y} & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} & 1+\frac{1}{x}+\frac{1}{xy}+\frac{1}{z}\\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{y} & 0 & 0 & 1+\frac{1}{z} & \frac{1}{x}+\frac{1}{z}
\end{pmatrix}$$

In[66]:=
```
sigmaCubicB4 = sigmaCubicB3\
   ~factorOut~{3, 8}~dropColumn~{5, 7}\
   ~cNot~{3, 5, x}(*eliminated cubic term*)\
   ~cNot~{5, 1, 1}~cNot~{8, 1, 1}~cNot~{7, 1, 1}\
   ~cNot~{7, 2, 1}~cNot~{6, 2, 1}\
   ~cNot~{3, 8, 1}~cNot~{2, 7, 1}~cNot~{6, 7, 1}~cNot~{8, 7, 1}\
   ~colOp~{4, 2, 1}~colOp~{3, 2, 1}~colOp~{1, 3, 1}(*only xy,yz in 1,2 and 7,8*)\
   ~cNot~{1, 6, y}~cNot~{1, 7, 1}~cNot~{1, 8, 1}\
   ~cNot~{5, 4, 1}~cNot~{4, 6, 1}(*only yz in 2; yz xy in 7,8*)\
   ~colOp~{3, 2, 1}~cNot~{8, 2, 1}~colOp~{2, 3, 1}~cNot~{2, 8, 1}\
   ~cNot~{3, 1, 1}\
   ~cNot~{1, 2, 1}~cNot~{2, 1, 1}~cNot~{1, 2, 1}\
   ~cNot~{8, 2, 1}~cNot~{2, 8, 1}~cNot~{8, 2, 1}\
   ~cNot~{7, 3, 1}~cNot~{3, 7, 1}~cNot~{7, 3, 1}\
   ~cNot~{7, 4, 1}~cNot~{4, 7, 1}~cNot~{7, 4, 1}\
   ~cNot~{4, 2, 1}~cNot~{4, 3, 1}~cNot~{8, 1, 1}~cNot~{8, 4, 1}\
   ~colOp~{2, 4, 1}~colOp~{4, 2, 1}~colOp~{2, 4, 1};
sigmaCubicB4 // display2
```

Out[67]//MatrixForm=

$$\begin{pmatrix}
1+x & x+z & 1+x+z+y\,z & 1+x & 0 & 0 & 0 & 0 \\
1+z & 1+x & 0 & x+y\,z & 0 & 0 & 0 & 0 \\
1+y & x+y & 1+x+x\,y+z & x+z & 0 & 0 & 0 & 0 \\
1+x & 1+y & 1+x+x\,y+z & z+y\,z & 0 & 0 & 0 & 0 \\
0 & 0 & 1+y & 1+z & 0 & 0 & 0 & 0 \\
0 & 0 & 1+z & 1+x & 0 & 0 & 0 & 0 \\
0 & 0 & y+z & 1+y & 0 & 0 & 0 & 0 \\
0 & 0 & x\,y+y\,z & 1+x+z+y\,z & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x} & 0 & \frac{1}{x}+\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{x}+\frac{1}{y} & 0 & 1+\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{z} & 0 & \frac{1}{x}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{x}+\frac{1}{z} & 0 & 1+\frac{1}{x} \\
0 & 0 & 0 & 0 & 1+\frac{1}{x\,y} & \frac{1}{x}+\frac{1}{z} & \frac{1}{z}+\frac{1}{x\,z} & \frac{1}{x^2}+\frac{1}{x\,z} \\
0 & 0 & 0 & 0 & 1+\frac{1}{y\,z} & \frac{1}{x}+\frac{1}{y^2}+\frac{1}{z}+\frac{1}{y\,z} & \frac{1}{y}+\frac{1}{z^2} & 1+\frac{1}{z^2}+\frac{1}{x\,z}+\frac{1}{y\,z} \\
0 & 0 & 0 & 0 & 1+\frac{1}{y\,z} & \frac{1}{x}+\frac{1}{y}+\frac{1}{x\,y}+\frac{1}{z} & 1+\frac{1}{x}+\frac{1}{z^2}+\frac{1}{z} & 1+\frac{1}{z^2}+\frac{1}{x\,z}+\frac{1}{y\,z} \\
0 & 0 & 0 & 0 & 1+\frac{1}{y} & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} & 1+\frac{1}{z} & \frac{1}{x}+\frac{1}{z}
\end{pmatrix}$$

In[68]:=
```
sigmaCubicB5 = sigmaCubicB4\
   ~colOp~{2, 1, 1}(*Groebner-like thing...*)\
   ~cNot~{1, 8, 1}\
   ~cNot~{8, 5, x + z}\
   ~cNot~{8, 6, 1};
sigmaCubicB5 // display2
```

Out[69]//MatrixForm=

$$\begin{pmatrix}
1+x & 1+z & 1+x+xy+z & z+yz & 0 & 0 & 0 & 0 \\
1+z & x+z & 0 & x+yz & 0 & 0 & 0 & 0 \\
1+y & 1+x & 1+x+xy+z & x+z & 0 & 0 & 0 & 0 \\
1+x & x+y & 1+x+xy+z & z+yz & 0 & 0 & 0 & 0 \\
0 & 0 & 1+y & 1+z & 0 & 0 & 0 & 0 \\
0 & 0 & 1+z & 1+x & 0 & 0 & 0 & 0 \\
0 & 0 & y+z & 1+y & 0 & 0 & 0 & 0 \\
0 & 0 & 1+x & x+xz+yz+z^2 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x} & 0 & \frac{1}{x}+\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{x}+\frac{1}{y} & 0 & 1+\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{z} & 0 & \frac{1}{x}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{x}+\frac{1}{z} & 0 & 1+\frac{1}{x} \\
0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{z}+\frac{1}{yz} & \frac{1}{x}+\frac{1}{xy}+\frac{1}{z^2}+\frac{1}{z}+\frac{1}{xz}+\frac{1}{yz} & \frac{1}{x}+\frac{1}{z^2} & \frac{1}{x^2}+\frac{1}{xy}+\frac{1}{z^2}+\frac{1}{yz} \\
0 & 0 & 0 & 0 & \frac{1}{y}+\frac{1}{yz} & \frac{1}{x}+\frac{1}{y^2}+\frac{1}{y}+\frac{1}{yz} & 1+\frac{1}{y}+\frac{1}{z^2}+\frac{1}{z} & 1+\frac{1}{y}+\frac{1}{z^2}+\frac{1}{z}+\frac{1}{xz}+\frac{1}{yz} \\
0 & 0 & 0 & 0 & 1+\frac{1}{yz} & \frac{1}{x}+\frac{1}{y}+\frac{1}{xy}+\frac{1}{z} & 1+\frac{1}{x}+\frac{1}{z^2}+\frac{1}{z} & 1+\frac{1}{z^2}+\frac{1}{xz}+\frac{1}{yz} \\
0 & 0 & 0 & 0 & 1+\frac{1}{y} & \frac{1}{y}+\frac{1}{z} & 1+\frac{1}{z} & \frac{1}{y}+\frac{1}{z}
\end{pmatrix}$$

In[70]:=
```
sigmaCubicB6 = sigmaCubicB5\
   ~cNot~{3, 1, 1}~cNot~{4, 1, 1}\
   ~cNot~{8, 5, z}~cNot~{8, 6, z}~cNot~{8, 7, z}\
   ~cNot~{1, 5, x}~cNot~{1, 6, 1}\
   ~colOp~{4, 2, 1}~cNot~{4, 5, 1}~cNot~{4, 6, 1}~cNot~{4, 7, 1}\
   ~cNot~{3, 2, 1}~PolynomialMod~2;
sigmaCubicB6 // display2
```

Out[71]//MatrixForm=

$$\begin{pmatrix}
1+x & 1+z & 0 & x\,z+y\,z & 0 & 0 & 0 & 0 \\
1+z & x+z & 0 & z+y\,z & 0 & 0 & 0 & 0 \\
1+x+y+z & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 1+x+y+z & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 1+y & 1+z & 0 & 0 & 0 & 0 \\
0 & 0 & 1+z & 1+x & 0 & 0 & 0 & 0 \\
0 & 0 & y+z & 1+y & 0 & 0 & 0 & 0 \\
0 & 0 & 1+x & x+z & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} & 0 & 1+\frac{1}{x}+\frac{1}{y}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & 1+\frac{1}{z} & 0 & \frac{1}{x}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & \frac{1}{x}+\frac{1}{z} & 0 & 1+\frac{1}{x} \\
0 & 0 & 0 & 0 & 1+\frac{1}{x} & \frac{1}{x\,y}+\frac{1}{x\,z} & \frac{1}{x}+\frac{1}{z} & 1+\frac{1}{x\,z} \\
0 & 0 & 0 & 0 & \frac{1}{y}+\frac{1}{z} & \frac{1}{y^2}+\frac{1}{y}+\frac{1}{z^2}+\frac{1}{z} & 1+\frac{1}{y} & 1+\frac{1}{x\,z} \\
0 & 0 & 0 & 0 & 1+\frac{1}{z} & \frac{1}{y}+\frac{1}{x\,y}+\frac{1}{z^2}+\frac{1}{y\,z} & 1+\frac{1}{x} & \frac{1}{x}+\frac{1}{x\,z} \\
0 & 0 & 0 & 0 & 1+\frac{1}{y} & \frac{1}{y}+\frac{1}{z} & 1+\frac{1}{z} & \frac{1}{y}+\frac{1}{z}
\end{pmatrix}$$

```
In[72]:= sigmaCubicB7 = sigmaCubicB6\
    ~cNot~{1, 6, z}~cNot~{1, 7, z}\
    ~cNot~{1, 5, z}~cNot~{2, 7, z}\
    ~colOp~{4, 2, z}~cNot~{4, 5, z}~cNot~{4, 6, z}~cNot~{4, 7, z}\
    ~cNot~{2, 5, z}~cNot~{2, 6, z}\
    ~colOp~{6, 7, 1 / y}~colOp~{6, 5, 1 / z}~colOp~{8, 7, 1}~colOp~{8, 5, 1}\
    ~colOp~{5, 8, 1}~colOp~{8, 5, 1}~colOp~{5, 8, 1}\
    ~colOp~{2, 1, 1}~colOp~{1, 2, 1}\
    ~cNot~{2, 1, 1}~cNot~{1, 2, 1}~cNot~{3, 1, 1}~cNot~{3, 2, 1}~cNot~{3, 4, 1}~cNot~{4, 1, 1}\
    ~cNot~{3, 2, 1}~cNot~{4, 2, 1}\
    ~cNot~{3, 4, 1}~cNot~{4, 3, 1}~cNot~{3, 4, 1}\
    ~colOp~{4, 3, 1}~cNot~{7, 6, 1}~cNot~{8, 6, 1}~cNot~{6, 8, 1}~cNot~{5, 6, 1}\
    ~cNot~{5, 8, 1}~cNot~{8, 5, 1}~cNot~{5, 8, 1}\
    ~cNot~{7, 8, 1}~cNot~{8, 7, 1}~cNot~{7, 8, 1}\
    ~colOp~{6, 5, 1}~colOp~{5, 6, 1}\
    ~colOp~{7, 8, 1}~colOp~{8, 7, 1}~colOp~{7, 8, 1};
  sigmaCubicB7 // display2
```

Out[73]//MatrixForm=

$$\begin{pmatrix}
x+z & 1+x & 0 & 0 & 0 & 0 & 0 & 0 \\
1+x & 1+z & 0 & 0 & 0 & 0 & 0 & 0 \\
x+y & 1+y & 0 & 0 & 0 & 0 & 0 & 0 \\
1+y & 1+x & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & x+z & 1+x & 0 & 0 & 0 & 0 \\
0 & 0 & 1+x & 1+z & 0 & 0 & 0 & 0 \\
0 & 0 & x+y & 1+y & 0 & 0 & 0 & 0 \\
0 & 0 & 1+y & 1+x & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{y} & \frac{1}{x}+\frac{1}{y} & 0 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{x} & 1+\frac{1}{y} & 0 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{x} & \frac{1}{x}+\frac{1}{z} & 0 & 0 \\
0 & 0 & 0 & 0 & 1+\frac{1}{z} & 1+\frac{1}{x} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{y} & \frac{1}{x}+\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x} & 1+\frac{1}{y} \\
0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{x} & \frac{1}{x}+\frac{1}{z} \\
0 & 0 & 0 & 0 & 0 & 0 & 1+\frac{1}{z} & 1+\frac{1}{x}
\end{pmatrix}$$

The new model bifurcates into two copies of itself.